```
$TITLE  M9_1.GMS: Two-Country Oligopoly, free entry, segmented markets
*
$ONTEXT
        YI    YJ     XI    XJ    NI    NJ    PUI   PUJ  CONI  CONJ  EHTI ENTJ

PYI    100                                 -100
PYJ          100                                -100
PXI                100                      -50   -50
PXJ                      100                -50   -50
FCI                            20                                  -20
FCJ                                  20                                -20
PSI    -40           -48          -12                 100
PSJ           -40          -48          -12                 100
PUI    -60           -32          -8                  100
PUJ           -60          -32          -8                  100
PWI                                       200         -200
PWI                                              200         -200
MKI                -10   -10                                      10    10
MKJ                -10   -10                                      10    10
$OFFTEXT

PARAMETERS
 TC          trade costs on a gross basis (TC = 1 is costless trade)
 FC          fixed costs
 ENDOWIS     endowment of skilled labor in country i
 ENDOWIL     endowment of unskilled labor in country i
```

```
 ENDOWJS     endowment of skilled labor in country j
 ENDOWJL     endowment of unskilled labor in country j
 SUBSIDY     subsidy to X productioni in country i
 MODELSTAT   indicator whether or not model solved
 REALPUI     real price of unskilled labor in i
 REALPUJ     real price of unskilled labor in j
 REALPSI     real price of skilled labor in i
 REALPSJ     real price of skilled labor in j
 XOPFI       X output per firm in country i
 XOPFJ       X output per firm in country j;

ENDOWIL = 1;
ENDOWIS = 1;
ENDOWJL = 1;
ENDOWJS = 1;
TC = 1;
FC = 8;
SUBSIDY = 0;

POSITIVE VARIABLES
 WFI         welfare in country i
 WFJ         welfare in country j
 YI          production of Y in i
 YJ          production of Y in j
 XI          production of X in i
 XII         supply of XI to market i
```

```
XIJ         supply of XI to market j  (XIJ shipped XIJ over TC recieved)
XJ          production of X in j
XJJ         supply of XJ to market j
XJI         supply of XJ to market j  (XIJ shipped XIJ over TC recieved)
NI          number of firms in (headquartered in) i
NJ          number of firms in (headquartered in) j
PY          domestic and world price of Y (no trade costs)
PWI         real consumer price index in i
PWJ         real consumer price index in j
PUI         price of unskilled labor in i
PUJ         price of unskilled labor in j
PSI         price of skilled labor in i
PSJ         price of skilled labor in j
PXI         price of X in i
PXJ         price of X in j
PXDI        producer marginal cost of X in i
PXDJ        producer marginal cost of X in j
PFI         price of fixed costs in i
PFJ         price of fixed costs in j
CONSI       consumer income in i
CONSJ       consumer income in j
ENTI        entrepreneurs' markup revenues in i
ENTJ        entrepreneurs' markup revenues in j
MARKII      markup on a firm from i's sales in i
MARKIJ      markup on a firm from i's sales in j
MARKJI      markup on a firm from j's sales in i
```

```
 MARKJJ      markup on a firm from j's sales in j;

EQUATIONS
PRWI         Zero profits for WFI
PRWJ         Zero profits for WFJ
PRXDI        Marginal cost of X in i
PRXII        MR = MC for XII
PRXIJ        MR = MC for XIJ
PRXDJ        Marginal cost of X in j
PRXJJ        MR = MC for XJJ
PRXJI        MR = MC for XJI
PRYI         Zero profits for YI
PRYJ         Zero profits for YJ
PRFI         Zero profits for FI
PRFJ         Zero profits for FJ
DXDI         X output in country i
DXI          Demand for X in country i
DXDJ         X output in country j
DXJ          Demand for X in country j
DY           Demand for Y
DWI          Demand for welfare in country i
DWJ          Demand for welfare in country j
DFI          Demand for fixed costs in i (markup revenues = fixed costs)
DFJ          Demand for fixed costs in j (markup revenues = fixed costs)
SKLABI       Market clearing for SI
SKLABJ       Market clearing for SJ
```

```
UNLABI      Market clearing for LI
UNLABJ      Market clearing for LJ
ICONSI      Consumer income in i
ICONSJ      Consumer income in j
IENTREI     Entreprenuer's income (markups) in i
IENTREJ     Entrepreneur's income (markups) in j
MKII        Markup ii
MKIJ        Markup ij
MKJJ        Markup jj
MKJI        Markup ji;


PRXDI..   (PUI**0.40)*(PSI**0.60)*(1-SUBSIDY) =G= PXDI;

PRXII..   PXDI =G= PXI*(1 - MARKII);

PRXIJ..   PXDI*TC =G= PXJ*(1 - MARKIJ);

PRXDJ..   (PUJ**0.40)*(PSJ**0.60) =G= PXDJ;

PRXJJ..   PXDJ =G= PXJ*(1 - MARKJJ);

PRXJI..   PXDJ*TC =G= PXI*(1 - MARKJI);

PRYI..   (PUI**0.60)*(PSI**0.40) =G= PY;
```

```
PRYJ..   (PUJ**0.60)*(PSJ**0.40) =G= PY;

PRWI..   ((PXI/1.25)**0.5)*(PY**0.5) =G= PWI;

PRWJ..   ((PXJ/1.25)**0.5)*(PY**0.5) =G= PWJ;

PRFI..   (PUI**0.40)*(PSI**0.60) =G= PFI;

PRFJ..   (PUJ**0.40)*(PSJ**0.60) =G= PFJ;

DXDI..   XII*40 + XIJ*40 =E= XI*80;
DXDJ..   XJJ*40 + XJI*40 =E= XJ*80;

DXI..    (XII*40 + XJI*40/TC) =E= 0.5*CONSI/PXI;
DXJ..    (XJJ*40 + XIJ*40/TC) =E= 0.5*CONSJ/PXJ;

DY..     (YI + YJ)*100 =E= 0.5*(CONSI + CONSJ)/PY;

DWI..    WFI*200 =E= CONSI/PWI;

DWJ..    WFJ*200 =E= CONSJ/PWJ;

DFI..    NI*FC =G= ENTI/PFI;

DFJ..    NJ*FC =G= ENTJ/PFJ;
```

```
SKLABI.. 100*ENDOWIS =E= 0.40*YI*100*PY/PSI
       + 0.60*(XII+XIJ)*40*(PXDI/(1-SUBSIDY))/PSI + 0.60*NI*FC*PFI/PSI;

SKLABJ.. 100*ENDOWJS =E= 0.40*YJ*100*PY/PSJ
       + 0.60*(XJJ+XJI)*40*PXDJ/PSJ  +  0.60*NJ*FC*PFJ/PSJ;

UNLABI.. 100*ENDOWIL =E= 0.60*YI*100*PY/PUI
       + 0.40*(XII+XIJ)*40*(PXDI/(1-SUBSIDY))/PUI + 0.40*NI*FC*PFI/PUI;

UNLABJ.. 100*ENDOWJL =E= 0.60*YJ*100*PY/PUJ
       + 0.40*(XJJ+XJI)*40*PXDJ/PUJ  +  0.40*NJ*FC*PFJ/PUJ;

ICONSI.. CONSI =E= PSI*100*ENDOWIS + PUI*100*ENDOWIL
                  -(PUI**0.40)*(PSI**0.60)*SUBSIDY*XI*80;

ICONSJ.. CONSJ =E= PSJ*100*ENDOWJS + PUJ*100*ENDOWJL;


IENTREI..  ENTI =G= MARKII*PXI*XII*40 + MARKIJ*PXJ*(XIJ/TC)*40;
IENTREJ..  ENTJ =G= MARKJJ*PXJ*XJJ*40 + MARKJI*PXI*(XJI/TC)*40;

MKII..     MARKII =E= (XII/NI)/(XII + XJI/TC);
MKIJ..     MARKIJ =E= (XIJ/TC/NI)/(XIJ/TC + XJJ);
MKJJ..     MARKJJ =E= (XJJ/NJ)/(XIJ/TC + XJJ);
MKJI..     MARKJI =E= (XJI/TC/NJ)/(XII + XJI/TC);
```

```
MODEL M9_1 /DXDI.PXDI, DXDJ.PXDJ, DXI.PXI, DXJ.PXJ, DY.PY,
           DWI.PWI, DWJ.PWJ, DFI.PFI, DFJ.PFJ,
           PRXDI.XI, PRXII.XII, PRXIJ.XIJ,
           PRXDJ.XJ, PRXJJ.XJJ, PRXJI.XJI,
           PRYI.YI, PRYJ.YJ, PRWI.WFI, PRWJ.WFJ,
           PRFI.NI, PRFJ.NJ, SKLABI.PSI, SKLABJ.PSJ,
           UNLABI.PUI, UNLABJ.PUJ, ICONSI.CONSI, ICONSJ.CONSJ,
           IENTREI.ENTI, IENTREJ.ENTJ,
           MKII.MARKII, MKIJ.MARKIJ, MKJJ.MARKJJ, MKJI.MARKJI/;

CONSI.L = 200;
CONSJ.L = 200;
ENTI.L = 20;
ENTJ.L = 20;
XI.L = 1;
XJ.L = 1;
XII.L = 1;
XIJ.L = 1;
XJJ.L = 1;
XJI.L = 1;
YI.L = 1;
YJ.L = 1;
WFI.L = 1;
WFJ.L = 1;
NI.L = 2.5;
```

```
NJ.L = 2.5;
PXDI.L = 1;
PXDJ.L = 1;
PXI.L = 1.25;
PXJ.L = 1.25;
PY.L = 1;
PSI.L = 1;
PSJ.L = 1;
PUI.L = 1;
PUJ.L = 1;
PWI.L = 1;
PWJ.L = 1;
PFI.L = 1;
PFJ.L = 1;
MARKII.L = 0.20;
MARKIJ.L = 0.20;
MARKJJ.L = 0.20;
MARKJI.L = 0.20;

PY.FX = 1;

*M9_1.ITERLIM = 0;
SOLVE M9_1 USING MCP;
MODELSTAT = M9_1.MODELSTAT - 1.;

* counterfactual: trade costs of 10%
```

```
TC = 1.1;

SOLVE M9_1 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;
XOPFI = XI.L/(NI.L/2.5);
XOPFJ = XJ.L/(NJ.L/2.5);

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ, XOPFI, XOPFJ;

* counterfactual: country's identical except for size,
* positive trade costs (home market advantage)

TC = 1.1;
ENDOWIL = 1.5;
ENDOWIS = 1.5;
ENDOWJL = 0.5;
ENDOWJS = 0.5;

SOLVE M9_1 USING MCP;

REALPUI = PUI.L/PWI.L;
```

```
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;
XOPFI = XI.L/(NI.L/2.5);
XOPFJ = XJ.L/(NJ.L/2.5);

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ, XOPFI, XOPFJ;

* counterfactual: country h has a comparative advantage in X

TC = 1.1;
ENDOWIL = 0.8;
ENDOWIS = 1.2;
ENDOWJL = 1.2;
ENDOWJS = 0.8;

SOLVE M9_1 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;
XOPFI = XI.L/(NI.L/2.5);
XOPFJ = XJ.L/(NJ.L/2.5);

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ, XOPFI, XOPFJ;
```

```
* counterfactual: country h has a comparative advantage in X
* no trade costs

TC = 1.;
ENDOWIL = 0.80;
ENDOWIS = 1.20;
ENDOWJL = 1.20;
ENDOWJS = 0.80;

SOLVE M9_1 USING MCP;

REALPUI = PUI.L/PWI.L;
REALPUJ = PUJ.L/PWJ.L;
REALPSI = PSI.L/PWI.L;
REALPSJ = PSJ.L/PWJ.L;
XOPFI = XI.L/(NI.L/2.5);
XOPFJ = XJ.L/(NJ.L/2.5);

DISPLAY REALPUI, REALPUJ, REALPSI, REALPSJ, XOPFI, XOPFJ;
```